// File: contracts/manifold/creator-core/core/CreatorCore.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/// @author: manifold.xyz

import "../../../openzeppelin/security/ReentrancyGuard.sol";
import "../../../openzeppelin/utils/Strings.sol";
import "../../../openzeppelin/utils/introspection/ERC165.sol";
import "../../../openzeppelin/utils/introspection/ERC165Checker.sol";
import "../../../openzeppelin/utils/structs/EnumerableSet.sol";
import "../../../openzeppelin-upgradeable/utils/AddressUpgradeable.sol";

import "../extensions/ICreatorExtensionTokenURI.sol";

import "./ICreatorCore.sol";

/**
 * @dev Core creator implementation
 */
abstract contract CreatorCore is ReentrancyGuard, ICreatorCore, ERC165 {
    using Strings for uint256;
    using EnumerableSet for EnumerableSet.AddressSet;
    using AddressUpgradeable for address;

    /// @custom:oz-upgrades-unsafe-allow state-variable-assignment
    uint256 _tokenCount = 0;

    // Track registered extensions data
    EnumerableSet.AddressSet internal _extensions;
    EnumerableSet.AddressSet internal _blacklistedExtensions;
    mapping(address => address) internal _extensionPermissions;
    mapping(address => bool) internal _extensionApproveTransfers;

    // For tracking which extension a token was minted by
    mapping(uint256 => address) internal _tokensExtension;

    // The baseURI for a given extension
    mapping(address => string) private _extensionBaseURI;
    mapping(address => bool) private _extensionBaseURIIdentical;

    // The prefix for any tokens with a uri configured
    mapping(address => string) private _extensionURIPrefix;

    // Mapping for individual token URIs
    mapping(uint256 => string) internal _tokenURIs;

    // Royalty configurations
    mapping(address => address payable[]) internal _extensionRoyaltyReceivers;
    mapping(address => uint256[]) internal _extensionRoyaltyBPS;
    mapping(uint256 => address payable[]) internal _tokenRoyaltyReceivers;
    mapping(uint256 => uint256[]) internal _tokenRoyaltyBPS;

    /**
     * External interface identifiers for royalties
     */

    /**
     *  @dev CreatorCore
     *
     *  bytes4(keccak256('getRoyalties(uint256)')) == 0xbb3bafd6
     *
     *  => 0xbb3bafd6 = 0xbb3bafd6
     */
    bytes4 private constant _INTERFACE_ID_ROYALTIES_CREATORCORE = 0xbb3bafd6;

    /**
     *  @dev Rarible: RoyaltiesV1
     *
     *  bytes4(keccak256('getFeeRecipients(uint256)')) == 0xb9c4d9fb
     *  bytes4(keccak256('getFeeBps(uint256)')) == 0x0ebd4c7f
     *
     *  => 0xb9c4d9fb ^ 0x0ebd4c7f = 0xb7799584
     */
    bytes4 private constant _INTERFACE_ID_ROYALTIES_RARIBLE = 0xb7799584;

    /**
     *  @dev Foundation
     *
     *  bytes4(keccak256('getFees(uint256)')) == 0xd5a06d4c
     *
     *  => 0xd5a06d4c = 0xd5a06d4c
     */
    bytes4 private constant _INTERFACE_ID_ROYALTIES_FOUNDATION = 0xd5a06d4c;

    /**
     *  @dev EIP-2981
     *
     * bytes4(keccak256("royaltyInfo(uint256,uint256)")) == 0x2a55205a
     *
     * => 0x2a55205a = 0x2a55205a
     */
    bytes4 private constant _INTERFACE_ID_ROYALTIES_EIP2981 = 0x2a55205a;

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId)
        public
        view
        virtual
        override(ERC165, IERC165)
        returns (bool)
    {
        return
            interfaceId == type(ICreatorCore).interfaceId ||
            super.supportsInterface(interfaceId) ||
            interfaceId == _INTERFACE_ID_ROYALTIES_CREATORCORE ||
            interfaceId == _INTERFACE_ID_ROYALTIES_RARIBLE ||
            interfaceId == _INTERFACE_ID_ROYALTIES_FOUNDATION ||
            interfaceId == _INTERFACE_ID_ROYALTIES_EIP2981;
    }

    /**
     * @dev Only allows registered extensions to call the specified function
     */
    modifier extensionRequired() {
        require(
            _extensions.contains(msg.sender),
            "Must be registered extension"
        );
        _;
    }

    /**
     * @dev Only allows non-blacklisted extensions
     */
    modifier nonBlacklistRequired(address extension) {
        require(
            !_blacklistedExtensions.contains(extension),
            "Extension blacklisted"
        );
        _;
    }

    /**
     * @dev See {ICreatorCore-getExtensions}.
     */
    function getExtensions()
        external
        view
        override
        returns (address[] memory extensions)
    {
        extensions = new address[](_extensions.length());
        for (uint256 i = 0; i < _extensions.length(); i++) {
            extensions[i] = _extensions.at(i);
        }
        return extensions;
    }

    /**
     * @dev Register an extension
     */
    function _registerExtension(
        address extension,
        string calldata baseURI,
        bool baseURIIdentical
    ) internal {
        require(extension != address(this), "Creator: Invalid");
        require(
            extension.isContract(),
            "Creator: Extension must be a contract"
        );
        if (!_extensions.contains(extension)) {
            _extensionBaseURI[extension] = baseURI;
            _extensionBaseURIIdentical[extension] = baseURIIdentical;
            emit ExtensionRegistered(extension, msg.sender);
            _extensions.add(extension);
        }
    }

    /**
     * @dev Unregister an extension
     */
    function _unregisterExtension(address extension) internal {
        if (_extensions.contains(extension)) {
            emit ExtensionUnregistered(extension, msg.sender);
            _extensions.remove(extension);
        }
    }

    /**
     * @dev Blacklist an extension
     */
    function _blacklistExtension(address extension) internal {
        require(extension != address(this), "Cannot blacklist yourself");
        if (_extensions.contains(extension)) {
            emit ExtensionUnregistered(extension, msg.sender);
            _extensions.remove(extension);
        }
        if (!_blacklistedExtensions.contains(extension)) {
            emit ExtensionBlacklisted(extension, msg.sender);
            _blacklistedExtensions.add(extension);
        }
    }

    /**
     * @dev Set base token uri for an extension
     */
    function _setBaseTokenURIExtension(string calldata uri, bool identical)
        internal
    {
        _extensionBaseURI[msg.sender] = uri;
        _extensionBaseURIIdentical[msg.sender] = identical;
    }

    /**
     * @dev Set token uri prefix for an extension
     */
    function _setTokenURIPrefixExtension(string calldata prefix) internal {
        _extensionURIPrefix[msg.sender] = prefix;
    }

    /**
     * @dev Set token uri for a token of an extension
     */
    function _setTokenURIExtension(uint256 tokenId, string calldata uri)
        internal
    {
        require(_tokensExtension[tokenId] == msg.sender, "Invalid token");
        _tokenURIs[tokenId] = uri;
    }

    /**
     * @dev Set base token uri for tokens with no extension
     */
    function _setBaseTokenURI(string memory uri) internal {
        _extensionBaseURI[address(this)] = uri;
    }

    /**
     * @dev Set token uri prefix for tokens with no extension
     */
    function _setTokenURIPrefix(string calldata prefix) internal {
        _extensionURIPrefix[address(this)] = prefix;
    }

    /**
     * @dev Set token uri for a token with no extension
     */
    function _setTokenURI(uint256 tokenId, string calldata uri) internal {
        require(_tokensExtension[tokenId] == address(this), "Invalid token");
        _tokenURIs[tokenId] = uri;
    }

    /**
     * @dev Retrieve a token's URI
     */
    function _tokenURI(uint256 tokenId) internal view returns (string memory) {
        address extension = _tokensExtension[tokenId];
        require(
            !_blacklistedExtensions.contains(extension),
            "Extension blacklisted"
        );

        if (bytes(_tokenURIs[tokenId]).length != 0) {
            if (bytes(_extensionURIPrefix[extension]).length != 0) {
                return
                    string(
                        abi.encodePacked(
                            _extensionURIPrefix[extension],
                            _tokenURIs[tokenId]
                        )
                    );
            }
            return _tokenURIs[tokenId];
        }

        if (
            ERC165Checker.supportsInterface(
                extension,
                type(ICreatorExtensionTokenURI).interfaceId
            )
        ) {
            return
                ICreatorExtensionTokenURI(extension).tokenURI(
                    address(this),
                    tokenId
                );
        }
        if (bytes(_extensionBaseURI[extension]).length != 0) {
            if (!_extensionBaseURIIdentical[extension]) {
                return
                    string(
                        abi.encodePacked(
                            _extensionBaseURI[extension],
                            tokenId.toString()
                        )
                    );
            } else {
                return _extensionBaseURI[extension];
            }
        } else {
            if (!_extensionBaseURIIdentical[address(this)])
            {
            return
                string(
                    abi.encodePacked(
                        _extensionBaseURI[address(this)],
                        tokenId.toString()
                    )
                );
            }else{
                return _extensionBaseURI[address(this)];
            }
        }
    }

    /**
     * Get token extension
     */
    function _tokenExtension(
        uint256 tokenId
    ) internal view returns (address extension) {
        extension = _tokensExtension[tokenId];

        require(extension != address(this), "No extension for token");
        require(
            !_blacklistedExtensions.contains(extension),
            "Extension blacklisted"
        );
        return extension;
    }

    /**
     * Helper to get royalties for a token
     */
    function _getRoyalties(
        uint256 tokenId
    ) internal view returns (address payable[] storage, uint256[] storage) {
        return (_getRoyaltyReceivers(tokenId), _getRoyaltyBPS(tokenId));
    }

    /**
     * Helper to get royalty receivers for a token
     */
    function _getRoyaltyReceivers(
        uint256 tokenId
    ) internal view returns (address payable[] storage) {
        if (_tokenRoyaltyReceivers[tokenId].length > 0) {
            return _tokenRoyaltyReceivers[tokenId];
        } else if (
            _extensionRoyaltyReceivers[_tokensExtension[tokenId]].length > 0
        ) {
            return _extensionRoyaltyReceivers[_tokensExtension[tokenId]];
        }
        return _extensionRoyaltyReceivers[address(this)];
    }

    /**
     * Helper to get royalty basis points for a token
     */
    function _getRoyaltyBPS(
        uint256 tokenId
    ) internal view returns (uint256[] storage) {
        if (_tokenRoyaltyBPS[tokenId].length > 0) {
            return _tokenRoyaltyBPS[tokenId];
        } else if (_extensionRoyaltyBPS[_tokensExtension[tokenId]].length > 0) {
            return _extensionRoyaltyBPS[_tokensExtension[tokenId]];
        }
        return _extensionRoyaltyBPS[address(this)];
    }

    function _getRoyaltyInfo(
        uint256 tokenId,
        uint256 value
    ) internal view returns (address receiver, uint256 amount) {
        address payable[] storage receivers = _getRoyaltyReceivers(tokenId);
        require(receivers.length <= 1, "More than 1 royalty receiver");

        if (receivers.length == 0) {
            return (address(this), 0);
        }
        return (receivers[0], (_getRoyaltyBPS(tokenId)[0] * value) / 10000);
    }

    /**
     * Set royalties for a token
     */
    function _setRoyalties(
        uint256 tokenId,
        address payable[] calldata receivers,
        uint256[] calldata basisPoints
    ) internal {
        require(receivers.length == basisPoints.length, "Invalid input");
        uint256 totalBasisPoints;
        for (uint256 i = 0; i < basisPoints.length; i++) {
            totalBasisPoints += basisPoints[i];
        }
        require(totalBasisPoints < 10000, "Invalid total royalties");
        _tokenRoyaltyReceivers[tokenId] = receivers;
        _tokenRoyaltyBPS[tokenId] = basisPoints;
        emit RoyaltiesUpdated(tokenId, receivers, basisPoints);
    }

    /**
     * Set royalties for all tokens of an extension
     */
    function _setRoyaltiesExtension(
        address extension,
        address payable[] calldata receivers,
        uint256[] calldata basisPoints
    ) internal {
        require(receivers.length == basisPoints.length, "Invalid input");
        uint256 totalBasisPoints;
        for (uint256 i = 0; i < basisPoints.length; i++) {
            totalBasisPoints += basisPoints[i];
        }
        require(totalBasisPoints < 10000, "Invalid total royalties");
        _extensionRoyaltyReceivers[extension] = receivers;
        _extensionRoyaltyBPS[extension] = basisPoints;
        if (extension == address(this)) {
            emit DefaultRoyaltiesUpdated(receivers, basisPoints);
        } else {
            emit ExtensionRoyaltiesUpdated(extension, receivers, basisPoints);
        }
    }
}

// File: contracts/manifold/creator-core/core/ICreatorCore.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/// @author: manifold.xyz

import "../../../openzeppelin/utils/introspection/IERC165.sol";

/**
 * @dev Core creator interface
 */
interface ICreatorCore is IERC165 {
    event ExtensionRegistered(
        address indexed extension,
        address indexed sender
    );
    event ExtensionUnregistered(
        address indexed extension,
        address indexed sender
    );
    event ExtensionBlacklisted(
        address indexed extension,
        address indexed sender
    );
    event MintPermissionsUpdated(
        address indexed extension,
        address indexed permissions,
        address indexed sender
    );
    event RoyaltiesUpdated(
        uint256 indexed tokenId,
        address payable[] receivers,
        uint256[] basisPoints
    );
    event DefaultRoyaltiesUpdated(
        address payable[] receivers,
        uint256[] basisPoints
    );
    event ExtensionRoyaltiesUpdated(
        address indexed extension,
        address payable[] receivers,
        uint256[] basisPoints
    );
    event ExtensionApproveTransferUpdated(
        address indexed extension,
        bool enabled
    );

    /**
     * @dev gets address of all extensions
     */
    function getExtensions() external view returns (address[] memory);

    /**
     * @dev add an extension.  Can only be called by contract owner or admin.
     * extension address must point to a contract implementing ICreatorExtension.
     * Returns True if newly added, False if already added.
     */
    function registerExtension(address extension, string calldata baseURI)
        external;

    /**
     * @dev add an extension.  Can only be called by contract owner or admin.
     * extension address must point to a contract implementing ICreatorExtension.
     * Returns True if newly added, False if already added.
     */
    function registerExtension(
        address extension,
        string calldata baseURI,
        bool baseURIIdentical
    ) external;

    /**
     * @dev add an extension.  Can only be called by contract owner or admin.
     * Returns True if removed, False if already removed.
     */
    function unregisterExtension(address extension) external;

    /**
     * @dev blacklist an extension.  Can only be called by contract owner or admin.
     * This function will destroy all ability to reference the metadata of any tokens created
     * by the specified extension. It will also unregister the extension if needed.
     * Returns True if removed, False if already removed.
     */
    function blacklistExtension(address extension) external;

    /**
     * @dev set the baseTokenURI of an extension.  Can only be called by extension.
     */
    function setBaseTokenURIExtension(string calldata uri) external;

    /**
     * @dev set the baseTokenURI of an extension.  Can only be called by extension.
     * For tokens with no uri configured, tokenURI will return "uri+tokenId"
     */
    function setBaseTokenURIExtension(string calldata uri, bool identical)
        external;

    /**
     * @dev set the common prefix of an extension.  Can only be called by extension.
     * If configured, and a token has a uri set, tokenURI will return "prefixURI+tokenURI"
     * Useful if you want to use ipfs/arweave
     */
    function setTokenURIPrefixExtension(string calldata prefix) external;

    /**
     * @dev set the tokenURI of a token extension.  Can only be called by extension that minted token.
     */
    function setTokenURIExtension(uint256 tokenId, string calldata uri)
        external;

    /**
     * @dev set the tokenURI of a token extension for multiple tokens.  Can only be called by extension that minted token.
     */
    function setTokenURIExtension(
        uint256[] memory tokenId,
        string[] calldata uri
    ) external;

    /**
     * @dev set the baseTokenURI for tokens with no extension.  Can only be called by owner/admin.
     * For tokens with no uri configured, tokenURI will return "uri+tokenId"
     */
    function setBaseTokenURI(string calldata uri) external;

    /**
     * @dev set the common prefix for tokens with no extension.  Can only be called by owner/admin.
     * If configured, and a token has a uri set, tokenURI will return "prefixURI+tokenURI"
     * Useful if you want to use ipfs/arweave
     */
    function setTokenURIPrefix(string calldata prefix) external;

    /**
     * @dev set the tokenURI of a token with no extension.  Can only be called by owner/admin.
     */
    function setTokenURI(uint256 tokenId, string calldata uri) external;

    /**
     * @dev set the tokenURI of multiple tokens with no extension.  Can only be called by owner/admin.
     */
    function setTokenURI(uint256[] memory tokenIds, string[] calldata uris)
        external;

    /**
     * @dev set a permissions contract for an extension.  Used to control minting.
     */
    function setMintPermissions(address extension, address permissions)
        external;

    /**
     * @dev Configure so transfers of tokens created by the caller (must be extension) gets approval
     * from the extension before transferring
     */
    function setApproveTransferExtension(bool enabled) external;

    /**
     * @dev get the extension of a given token
     */
    function tokenExtension(uint256 tokenId) external view returns (address);

    /**
     * @dev Set default royalties
     */
    function setRoyalties(
        address payable[] calldata receivers,
        uint256[] calldata basisPoints
    ) external;

    /**
     * @dev Set royalties of a token
     */
    function setRoyalties(
        uint256 tokenId,
        address payable[] calldata receivers,
        uint256[] calldata basisPoints
    ) external;

    /**
     * @dev Set royalties of an extension
     */
    function setRoyaltiesExtension(
        address extension,
        address payable[] calldata receivers,
        uint256[] calldata basisPoints
    ) external;

    /**
     * @dev Get royalites of a token.  Returns list of receivers and basisPoints
     */
    function getRoyalties(uint256 tokenId)
        external
        view
        returns (address payable[] memory, uint256[] memory);

    // Royalty support for various other standards
    function getFeeRecipients(uint256 tokenId)
        external
        view
        returns (address payable[] memory);

    function getFeeBps(uint256 tokenId)
        external
        view
        returns (uint256[] memory);

    function getFees(uint256 tokenId)
        external
        view
        returns (address payable[] memory, uint256[] memory);

    function royaltyInfo(uint256 tokenId, uint256 value)
        external
        view
        returns (address, uint256);
}

// File: contracts/manifold/creator-core/core/IERC1155CreatorCore.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/// @author: manifold.xyz

import "./CreatorCore.sol";

/**
 * @dev Core ERC1155 creator interface
 */
interface IERC1155CreatorCore is ICreatorCore {

    /**
     * @dev mint a token with no extension. Can only be called by an admin.
     *
     * @param to       - Can be a single element array (all tokens go to same address) or multi-element array (single token to many recipients)
     * @param amounts  - Can be a single element array (all recipients get the same amount) or a multi-element array
     * @param uris     - If no elements, all tokens use the default uri.
     *                   If any element is an empty string, the corresponding token uses the default uri.
     *
     *
     * Requirements: If to is a multi-element array, then uris must be empty or single element array
     *               If to is a multi-element array, then amounts must be a single element array or a multi-element array of the same size
     *               If to is a single element array, uris must be empty or the same length as amounts
     *
     * Examples:
     *    mintBaseNew(['0x....1', '0x....2'], [1], [])
     *        Mints a single new token, and gives 1 each to '0x....1' and '0x....2'.  Token uses default uri.
     *    
     *    mintBaseNew(['0x....1', '0x....2'], [1, 2], [])
     *        Mints a single new token, and gives 1 to '0x....1' and 2 to '0x....2'.  Token uses default uri.
     *    
     *    mintBaseNew(['0x....1'], [1, 2], ["", "http://token2.com"])
     *        Mints two new tokens to '0x....1'. 1 of the first token, 2 of the second.  1st token uses default uri, second uses "http://token2.com".
     *    
     * @return Returns list of tokenIds minted
     */
    function mintBaseNew(address[] calldata to, uint256[] calldata amounts, string[] calldata uris) external returns (uint256[] memory);

    /**
     * @dev batch mint existing token with no extension. Can only be called by an admin.
     *
     * @param to        - Can be a single element array (all tokens go to same address) or multi-element array (single token to many recipients)
     * @param tokenIds  - Can be a single element array (all recipients get the same token) or a multi-element array
     * @param amounts   - Can be a single element array (all recipients get the same amount) or a multi-element array
     *
     * Requirements: If any of the parameters are multi-element arrays, they need to be the same length as other multi-element arrays
     *
     * Examples:
     *    mintBaseExisting(['0x....1', '0x....2'], [1], [10])
     *        Mints 10 of tokenId 1 to each of '0x....1' and '0x....2'.
     *    
     *    mintBaseExisting(['0x....1', '0x....2'], [1, 2], [10, 20])
     *        Mints 10 of tokenId 1 to '0x....1' and 20 of tokenId 2 to '0x....2'.
     *    
     *    mintBaseExisting(['0x....1'], [1, 2], [10, 20])
     *        Mints 10 of tokenId 1 and 20 of tokenId 2 to '0x....1'.
     *    
     *    mintBaseExisting(['0x....1', '0x....2'], [1], [10, 20])
     *        Mints 10 of tokenId 1 to '0x....1' and 20 of tokenId 1 to '0x....2'.
     *    
     */
    function mintBaseExisting(address[] calldata to, uint256[] calldata tokenIds, uint256[] calldata amounts) external;

    /**
     * @dev mint a token from an extension. Can only be called by a registered extension.
     *
     * @param to       - Can be a single element array (all tokens go to same address) or multi-element array (single token to many recipients)
     * @param amounts  - Can be a single element array (all recipients get the same amount) or a multi-element array
     * @param uris     - If no elements, all tokens use the default uri.
     *                   If any element is an empty string, the corresponding token uses the default uri.
     *
     *
     * Requirements: If to is a multi-element array, then uris must be empty or single element array
     *               If to is a multi-element array, then amounts must be a single element array or a multi-element array of the same size
     *               If to is a single element array, uris must be empty or the same length as amounts
     *
     * Examples:
     *    mintExtensionNew(['0x....1', '0x....2'], [1], [])
     *        Mints a single new token, and gives 1 each to '0x....1' and '0x....2'.  Token uses default uri.
     *    
     *    mintExtensionNew(['0x....1', '0x....2'], [1, 2], [])
     *        Mints a single new token, and gives 1 to '0x....1' and 2 to '0x....2'.  Token uses default uri.
     *    
     *    mintExtensionNew(['0x....1'], [1, 2], ["", "http://token2.com"])
     *        Mints two new tokens to '0x....1'. 1 of the first token, 2 of the second.  1st token uses default uri, second uses "http://token2.com".
     *    
     * @return Returns list of tokenIds minted
     */
    function mintExtensionNew(address[] calldata to, uint256[] calldata amounts, string[] calldata uris) external returns (uint256[] memory);

    /**
     * @dev batch mint existing token from extension. Can only be called by a registered extension.
     *
     * @param to        - Can be a single element array (all tokens go to same address) or multi-element array (single token to many recipients)
     * @param tokenIds  - Can be a single element array (all recipients get the same token) or a multi-element array
     * @param amounts   - Can be a single element array (all recipients get the same amount) or a multi-element array
     *
     * Requirements: If any of the parameters are multi-element arrays, they need to be the same length as other multi-element arrays
     *
     * Examples:
     *    mintExtensionExisting(['0x....1', '0x....2'], [1], [10])
     *        Mints 10 of tokenId 1 to each of '0x....1' and '0x....2'.
     *    
     *    mintExtensionExisting(['0x....1', '0x....2'], [1, 2], [10, 20])
     *        Mints 10 of tokenId 1 to '0x....1' and 20 of tokenId 2 to '0x....2'.
     *    
     *    mintExtensionExisting(['0x....1'], [1, 2], [10, 20])
     *        Mints 10 of tokenId 1 and 20 of tokenId 2 to '0x....1'.
     *    
     *    mintExtensionExisting(['0x....1', '0x....2'], [1], [10, 20])
     *        Mints 10 of tokenId 1 to '0x....1' and 20 of tokenId 1 to '0x....2'.
     *    
     */
    function mintExtensionExisting(address[] calldata to, uint256[] calldata tokenIds, uint256[] calldata amounts) external;

    /**
     * @dev burn tokens. Can only be called by token owner or approved address.
     * On burn, calls back to the registered extension's onBurn method
     */
    function burn(address account, uint256[] calldata tokenIds, uint256[] calldata amounts) external;

    /**
     * @dev Total amount of tokens in with a given tokenId.
     */
    function totalSupply(uint256 tokenId) external view returns (uint256);
}
// File: contracts/manifold/creator-core/core/IERC721CreatorCore.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/// @author: manifold.xyz

import "./ICreatorCore.sol";

/**
 * @dev Core ERC721 creator interface
 */
interface IERC721CreatorCore is ICreatorCore {

    /**
     * @dev mint a token with no extension. Can only be called by an admin.
     * Returns tokenId minted
     */
    function mintBase(address to) external returns (uint256);

    /**
     * @dev mint a token with no extension. Can only be called by an admin.
     * Returns tokenId minted
     */
    function mintBase(address to, string calldata uri) external returns (uint256);

    /**
     * @dev batch mint a token with no extension. Can only be called by an admin.
     * Returns tokenId minted
     */
    function mintBaseBatch(address to, uint16 count) external returns (uint256[] memory);

    /**
     * @dev batch mint a token with no extension. Can only be called by an admin.
     * Returns tokenId minted
     */
    function mintBaseBatch(address to, string[] calldata uris) external returns (uint256[] memory);

    /**
     * @dev mint a token. Can only be called by a registered extension.
     * Returns tokenId minted
     */
    function mintExtension(address to) external returns (uint256);

    /**
     * @dev mint a token. Can only be called by a registered extension.
     * Returns tokenId minted
     */
    function mintExtension(address to, string calldata uri) external returns (uint256);

    /**
     * @dev batch mint a token. Can only be called by a registered extension.
     * Returns tokenIds minted
     */
    function mintExtensionBatch(address to, uint16 count) external returns (uint256[] memory);

    /**
     * @dev batch mint a token. Can only be called by a registered extension.
     * Returns tokenId minted
     */
    function mintExtensionBatch(address to, string[] calldata uris) external returns (uint256[] memory);

    /**
     * @dev burn a token. Can only be called by token owner or approved address.
     * On burn, calls back to the registered extension's onBurn method
     */
    function burn(uint256 tokenId) external;

}
// File: contracts/manifold/creator-core/extensions/ICreatorExtensionTokenURI.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/// @author: manifold.xyz

import "../../../openzeppelin/utils/introspection/IERC165.sol";

/**
 * @dev Implement this if you want your extension to have overloadable URI's
 */
interface ICreatorExtensionTokenURI is IERC165 {
    /**
     * Get the uri for a given creator/tokenId
     */
    function tokenURI(address creator, uint256 tokenId)
        external
        view
        returns (string memory);
}

// File: contracts/manifold/libraries-solidity/access/AdminControl.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/// @author: manifold.xyz

import "../../../openzeppelin/utils/introspection/ERC165.sol";
import "../../../openzeppelin/utils/structs/EnumerableSet.sol";
import "../../../openzeppelin/access/Ownable.sol";
import "./IAdminControl.sol";

abstract contract AdminControl is Ownable, IAdminControl, ERC165 {
    using EnumerableSet for EnumerableSet.AddressSet;

    // Track registered admins
    EnumerableSet.AddressSet private _admins;

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId)
        public
        view
        virtual
        override(ERC165, IERC165)
        returns (bool)
    {
        return
            interfaceId == type(IAdminControl).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev Only allows approved admins to call the specified function
     */
    modifier adminRequired() {
        require(
            owner() == msg.sender || _admins.contains(msg.sender),
            "AdminControl: Must be owner or admin"
        );
        _;
    }

    /**
     * @dev See {IAdminControl-getAdmins}.
     */
    function getAdmins()
        external
        view
        override
        returns (address[] memory admins)
    {
        admins = new address[](_admins.length());
        for (uint256 i = 0; i < _admins.length(); i++) {
            admins[i] = _admins.at(i);
        }
        return admins;
    }

    /**
     * @dev See {IAdminControl-approveAdmin}.
     */
    function approveAdmin(address admin) external override onlyOwner {
        if (!_admins.contains(admin)) {
            emit AdminApproved(admin, msg.sender);
            _admins.add(admin);
        }
    }

    /**
     * @dev See {IAdminControl-revokeAdmin}.
     */
    function revokeAdmin(address admin) external override onlyOwner {
        if (_admins.contains(admin)) {
            emit AdminRevoked(admin, msg.sender);
            _admins.remove(admin);
        }
    }

    /**
     * @dev See {IAdminControl-isAdmin}.
     */
    function isAdmin(address admin) public view override returns (bool) {
        return (owner() == admin || _admins.contains(admin));
    }
}

// File: contracts/manifold/libraries-solidity/access/IAdminControl.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/// @author: manifold.xyz

import "../../../openzeppelin/utils/introspection/IERC165.sol";

/**
 * @dev Interface for admin control
 */
interface IAdminControl is IERC165 {
    event AdminApproved(address indexed account, address indexed sender);
    event AdminRevoked(address indexed account, address indexed sender);

    /**
     * @dev gets address of all admins
     */
    function getAdmins() external view returns (address[] memory);

    /**
     * @dev add an admin.  Can only be called by contract owner.
     */
    function approveAdmin(address admin) external;

    /**
     * @dev remove an admin.  Can only be called by contract owner.
     */
    function revokeAdmin(address admin) external;

    /**
     * @dev checks whether or not given address is an admin
     * Returns True if they are
     */
    function isAdmin(address admin) external view returns (bool);
}

// File: contracts/mojito/interfaces/IDutchAuction.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity ^0.8.0;
pragma experimental ABIEncoderV2;

/**
 * @title Interface for Dutch Auction
 */
interface IDutchAuction {
    struct NftContractList {
        // The NFT contract of Erc-721/1155 for Sale
        address nftContractAddress;
        // The From tokenId of the nft for sale
        uint256 startTokenId;
        // The To tokenId of the nft for sale
        uint256 endTokenId;
        // The quantity of each token If Erc-1155 collection
        uint256 unit1155;
        // The address of the token owner
        address tokenOwner;
        // total number of nfts for sale
        uint64 noOfTokens;
        // The limit which an each address is allowed
        uint64 walletLimit;
        // The limit for each transaction
        uint64 txnLimit;
        // To specify for mint on Demand
        bool isMint;
        // the max token id randge to mint
        int256 maxTokenIDRange;
        // To specify nft for Instant Delivery
        bool isInstantDeliver;
        // To be Specified for allowing rebate
        bool isRebate;
        // To be Specified for allowing discount
        bool isDiscountRequired;
        // To specify for AuthSign for buy
        bool isSignerRequired;
        // select the type of sale
        TypeOfSale saleType;
        // The root hash of the whitelisted buyers
        bytes32 blacklistedBuyers;
    }

    enum TypeOfSale {
        // refers for sale type of Erc-721 collections
        erc_721_nft_type,
        // refers for sale type of Erc-1155 collections with multiple Quantity
        single_tokens_multiple_Quantity,
        // refers for sale type of Erc-1155 collections with multiple tokens with Same Quantity
        multiple_tokens_with_Same_Quantity
    }

    struct DutchAuctionList {
        // details of the nft collection list
        NftContractList collectionList;
        // The starting price of the sale
        uint128 startingPrice;
        // The minimum price of the sale
        uint128 reservedPrice;
        // The price to be reduced in certain time interval
        uint128 reducePrice;
        // The time of the sale starts
        uint32 auctionStartTime;
        // The time of the sale ends
        uint32 auctionEndTime;
        // The time interval of the price decrement
        uint32 timeForPriceDecrement;
        // The start time for price decrement to half of previous price
        uint32 halfLifeTime;
        // The allowed Payment Currency for the sale
        address paymentCurrency;
        // The list of settlement Address and Settlement Price
        SettlementList payoutList;
    }

    struct SettlementList {
        // The settlement address for the listed tokens
        address payable paymentSettlementAddress;
        // The tax settlement address for settlement of tax fee
        address payable taxSettlementAddress;
        // The commission address for settlement of commission fee
        address payable commissionAddress;
        // The platform address for settlement of platform fee
        address payable platformSettlementAddress;
        // The commission fee given in basis points
        uint16 commissionFeePercentage;
        // The platform fee given in basis points
        uint16 platformFeePercentage;
    }

    struct BuyerDetails {
        // The price of the NFTs while buying
        uint256[] price;
        // The quatity of the nfts
        uint256[] quantity;
        // The tax for price
        uint256[] tax;
        // The discount price of the nfts
        uint256[] priceDiscount;
        // The nft quantity bought at the discount price
        uint256[] quantityDiscount;
        // The tax for discount price
        uint256[] taxDiscount;
        // The Total tokens purchased by seller
        uint256 totalCap;
        // The rebate claimed status
        bool rebateClaimed;
    }

    struct LatestSaleDetails {
        // The price of the latest sale
        uint256 finalPrice;
        // The id of the next token to buy
        uint256 nextTokenId;
        // The total number of tokens purchased
        uint256 currentSupply;
        // The total number of tokens purchased with full discount
        uint256 fullDiscountSupply;
        // The total number of tokens purchased with zero discount
        uint256 zeroDiscountSupply;
        // represents that auction is ended or not
        bool auctionEnded;
        // represents that maxId reached
        bool isReachedMaxId;
    }

    struct Discount {
        // The discount percentage
        uint16 discountPercentage;
        // The expire timeof discount signature
        uint32 expirationTime;
        // The nonce value
        string nonce;
        // The signature generated by admin
        bytes signature;
        // The signer of the signature.
        address signer;
    }

    struct BuyList {
        // The buy amount to purchase NFT(s)
        uint256 price;
        // The tax amount of the purchase
        uint256 tax;
        // The number of tokens to be purchased
        uint256 quantity;
        // The quantity for 1155 if ERC1155
        uint256 unit1155;
        // The merkle tree path of buyer address
        bytes32[] blacklistedProof;
        // the discount information
        IDutchAuction.Discount discount;
    }

    function usedAuctionId(string memory dutchAuctionID)
    external view returns (bool);

    /**
     * @notice checks the address is admin address ornot
     * @param walletAddress the id of the created auction
     */
    function isAdmin(address walletAddress) external view returns (bool);

    /**
     * @notice checks the signature is used already or not
     * @param signature the discount signature
     */
    function discountUsed(bytes memory signature) external view returns (bool);

    /**
     * @notice gets the detils of the listed sale
     * @param dutchAuctionID the id of the created auction
     */
    function getListings(
        string memory dutchAuctionID
    ) external view returns (DutchAuctionList memory auctionist);

    /**
     * @notice gets the detils collection details of listed sale
     * @param dutchAuctionID the id of the created auction
     */
    function getLatestSaleInfo(
        string memory dutchAuctionID
    ) external view returns (LatestSaleDetails memory saleDetails);

    /**
     * @notice gets the buyers sale info of the listed sale
     * @param dutchAuctionID the id of the created auction
     * @param buyer the address of the buyer
     */
    function getBuyerInfo(
        string memory dutchAuctionID,
        address buyer
    ) external view returns (BuyerDetails memory buyerDetails);

    /**
     * @notice gets the list of all the buyers in the sale
     * @param dutchAuctionID the id of the created auction
     */
    function getBuyerAddress(
        string memory dutchAuctionID
    ) external view returns (address[] memory buyers);

    /**
     * @notice validates auction data before auction creation and updation
     * @param auctionList the List of the auction data
     */
    function __beforeCreateAuction(
        DutchAuctionList memory auctionList
    ) external view;

    /**
     * @notice validates tha buy data before buy in dutch Auction
     * @param dutchAuctionID the id of the created auction
     */
    function __buyValidation(
        string memory dutchAuctionID,
        address buyer,
        IDutchAuction.BuyList memory buyList
    ) external view returns (uint256, uint256);

    /**
     * @notice gets the current dutch auction price
     * @param dutchAuctionId the id of the created auction
     */
    function getCurrentDutchPrice(
        string memory dutchAuctionId
    ) external view returns (uint256 currentPrice);

    /**
     * @notice gets the claimable amount in rebate type auction
     * @param dutchAuctionID the id of the created auction
     * @param buyer the address of buyer wallet
     */
    function claimable(
        string memory dutchAuctionID,
        address buyer
    ) external view returns (uint256 totalClaimable, uint256 totalTaxClaimable);

    /**
     * @notice gets the settlement amount and tax for endAuction()
     * @param dutchAuctionID the id of the created auction
     */
    function computeSettlementAmountAndTax(
        string memory dutchAuctionID
    ) external returns (uint256, uint256);
}

// File: contracts/mojito/interfaces/IRoyaltyEngine.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity ^0.8.0;
pragma experimental ABIEncoderV2;

/**
 * @title Interface for RoyaltyEngine
 */
interface IRoyaltyEngine {
    /**
     * @notice Emits when an collection level Royalty is configured
     * @param collectionAddress contract address 
     * @param receivers set of royalty receivers
     * @param basisPoints set of royalty Bps
     */
    event RoyaltiesUpdated(
        address indexed collectionAddress,
        address payable[] receivers,
        uint256[] basisPoints
    );

    /**
     * @notice Emits when an Token level Royalty is configured
     * @param collectionAddress contract address 
     * @param tokenId Token Id 
     * @param receivers set of royalty receivers
     * @param basisPoints set of royalty Bps
     */
    event TokenRoyaltiesUpdated(
        address collectionAddress,
        uint256 indexed tokenId,
        address payable[] receivers,
        uint256[] basisPoints
    );
    
    /**
     * @notice Emits when address is added into Black List.
     * @param account BlackListed NFT contract address or wallet address
     * @param sender caller address
    **/
    event AddedBlacklistedAddress(
        address indexed account,
        address indexed sender
    );

    /**
     * @notice Eits when address is removed from Black List.
     * @param account BlackListed NFT contract address or wallet address
     * @param sender caller address
    **/
    event RevokedBlacklistedAddress(
        address indexed account,
        address indexed sender
    );
    
    /**
     * @notice Setting royalty for NFT Collection.
     * @param collectionAddress NFT contract address 
     * @param receivers set of royalty receivers
     * @param basisPoints set of royalty Bps
     */
    function setRoyalty(
        address collectionAddress,
        address payable[] calldata receivers,
        uint256[] calldata basisPoints
    ) external;
    
    /**
     * @notice Setting royalty for token.
     * @param collectionAddress contract address 
     * @param tokenId Token Id 
     * @param receivers set of royalty receivers
     * @param basisPoints set of royalty Bps
     */
    function setTokenRoyalty(
        address collectionAddress,
        uint256 tokenId,
        address payable[] calldata receivers,
        uint256[] calldata basisPoints
    ) external;

    /**
     * @notice getting collection level royalty information 
     * @param collectionAddress contract address 
     * @return receivers returns set of royalty receivers address
     * @return basisPoints returns set of Bps to calculate Shares.
    **/
    function getCollectionRoyalty(address collectionAddress)
    external view
    returns (address payable[] memory receivers, uint256[] memory basisPoints);

    /**
     * @notice getting royalty information from Other royalty standard.
     * @param collectionAddress contract address 
     * @param tokenId Token Id 
     * @return receivers returns set of royalty receivers address
     * @return basisPoints returns set of Bps to calculate Shares.
    **/
    function getRoyalty(address collectionAddress, uint256 tokenId)
        external
	view
        returns (address payable[] memory receivers, uint256[] memory basisPoints);
    
    /**
     * @notice Compute royalty Shares
     * @param collectionAddress contract address 
     * @param tokenId Token Id 
     * @param amount amount involved to compute the Shares. 
     * @return receivers returns set of royalty receivers address
     * @return basisPoints returns set of Bps.
     * @return feeAmount returns set of Shares.
    **/
    function getRoyaltySplitshare(
        address collectionAddress,
        uint256 tokenId,
        uint256 amount
    )
        external
	view
        returns (
            address payable[] memory receivers,
            uint256[] memory basisPoints,
            uint256[] memory feeAmount
        );
    
    /**
     * @notice Adds address as blacklist
     * @param commonAddress user wallet address 
    **/
    function blacklistAddress(address commonAddress) external;

    /**
     * @notice revoke the blacklistedAddress
     * @param commonAddress address info
    **/
    function revokeBlacklistedAddress(address commonAddress) external;
        
    /**
     * @notice checks the blacklistedAddress
     * @param commonAddress address info
    **/
    function isBlacklistedAddress(address commonAddress)
        external
        view
        returns (bool);
}
// File: contracts/mojito/onchain/DutchAuction.sol



// SPDX-License-Identifier: GPL-3.0
pragma solidity ^0.8.0;
pragma experimental ABIEncoderV2;

import {IERC721, IERC165} from "../../openzeppelin/token/ERC721/IERC721.sol";
import {IERC1155} from "../../openzeppelin/token/ERC1155/IERC1155.sol";
import {ReentrancyGuard} from "../../openzeppelin/security/ReentrancyGuard.sol";
import {IERC20} from "../../openzeppelin/token/ERC20/IERC20.sol";
import {SafeERC20} from "../../openzeppelin/token/ERC20/utils/SafeERC20.sol";
import {AdminControl} from "../../manifold/libraries-solidity/access/AdminControl.sol";
import {Address} from "../../openzeppelin/utils/Address.sol";
import {IERC721CreatorCore} from "../../manifold/creator-core/core/IERC721CreatorCore.sol";
import {IERC1155CreatorCore} from "../../manifold/creator-core/core/IERC1155CreatorCore.sol";
import "../../mojito/interfaces/IRoyaltyEngine.sol";
import "../../openzeppelin/utils/structs/EnumerableSet.sol";
import {IDutchAuction} from "../../mojito/interfaces/IDutchAuction.sol";

/**
 * @title DutchAuction
 * @dev This contract enables collectors and curators to run their own auctions.It is a unique type of auction
 * where the price of an item starts high and gradually decreases until it reachs the reserved price
 *
 */
contract DutchAuction is ReentrancyGuard, AdminControl {
    using SafeERC20 for IERC20;
    using EnumerableSet for EnumerableSet.AddressSet;

    /// @notice details of the buyer
    /// @param price the price of the NFTs
    /// @param quantity the nft quantity purchased at actual price
    /// @param tax the tax rate
    /// @param priceDiscount the discount price of the nfts
    /// @param quantityDiscount the nft quantity purchased at the discount price
    /// @param taxDiscount the tax for discount price
    /// @param totalCap the total nfts brought by the buyer
    /// @param rebateClaimed represents that rebate is claimed or not
    struct BuyerDetails {
        uint256[] price;
        uint256[] quantity;
        uint256[] tax;
        uint256[] priceDiscount;
        uint256[] quantityDiscount;
        uint256[] taxDiscount;
        uint256 totalCap;
        bool rebateClaimed;
    }

    /// @notice LatestSaleDetails of the particular auction id
    /// @param finalPrice the price of the latest buy
    /// @param nextTokenId the id of the next token to buy
    /// @param currentSupply the total number of tokens purchased by all buyers
    /// @param fullDiscountSupply the total number of tokens purchased with full discount
    /// @param zeroDiscountSupply the total number of tokens purchased with zero discount
    /// @param auctionEnded represents that auction is ended or not
    /// @param isReachedMaxId represents that minting reacheed the maxId 
    struct LatestSaleDetails {
        uint256 finalPrice;
        uint256 nextTokenId;
        uint256 currentSupply;
        uint256 fullDiscountSupply;
        uint256 zeroDiscountSupply;
        bool auctionEnded;
        bool isReachedMaxId;
    }

    /// @notice emits an event when the auction is created or updated.
    /// @param auctionId contains the id of the created sale
    /// @param list contains the details of auction created
    /// @param createdOrUpdated the details provide whether auction is created or updated
    event AuctionCreatedorUpdated(
        string auctionId,
        IDutchAuction.DutchAuctionList list,
        string createdOrUpdated
    );

    /// @notice emits an event when the NFT is purchased.
    /// @param auctionId the id of the created auction
    /// @param quantity the number of tokens purchased
    /// @param unit1155 the uints if ERC1155
    /// @param currentPrice the current price of the single token
    /// @param price the price amount of the purchase
    /// @param tax the tax amount of the purchase
    /// @param nonce the nonce used for the signature
    /// @param buyer the buyer address
    /// @param tokenIds the token ids brought by the buyer
    event BuyExecuted(
        string auctionId,
        uint256 quantity,
        uint256 unit1155,
        uint256 currentPrice,
        uint256 price,
        uint256 tax,
        string nonce,
        address buyer,
        uint256[] tokenIds
    );

    /// @notice emits an event when auction is ended
    /// @param auctionId the id of the created auction
    /// @param soldQuantity the total number of tokens sold
    /// @param finalPrice the final price of the NFT
    /// @param tokenIds the token ids brought by the buyer
    event AuctionEnded(
        string auctionId,
        uint256 soldQuantity,
        uint256 finalPrice,
        uint256[] tokenIds
    );

    /// @notice emits an event when the rebate is claimed
    /// @param auctionId the id of the created sale
    /// @param buyer the buyer address where the rebate is paid
    /// @param rebateAmount the rebate amount paid to the buyer
    /// @param rebateTax the rebate tax paid to the buyer
    /// @param paymentCurrency the currency the rebate is paid at
    event Rebated(
        string auctionId,
        address buyer,
        uint256 rebateAmount,
        uint256 rebateTax,
        address paymentCurrency
    );

    /// @notice emits an event when the auction is cancelled
    /// @param auctionId contains the id of the created auction
    event CancelAuction(string auctionId);

    /// @notice emits an event when an royalty payout is executed
    /// @param tokenContract the NFT contract address
    /// @param recipient the address of the royalty recipient(s)
    /// @param shares the shares sent to the royalty recipient(s)
    event RoyaltyPayout(
        address tokenContract,
        address recipient,
        uint256 shares
    );

    /// @notice emits an event when EndTime is updated
    /// @param auctionId contains the id of the created auction
    /// @param endTimeNew The new EndTime
    event EndTimeUpdated(string auctionId, uint256 endTimeNew);

    /// @notice emits an event when Reserved Price is updated
    /// @param auctionId contains the id of the created auction
    /// @param reservePriceNew The new reserved price address
    event ReservePriceUpdated(string auctionId, uint256 reservePriceNew);

    /// @notice emits an event when an utility And royalty address are updated.
    /// @param dutchUtilityNew The new DutchUtility address
    /// @param royaltySupporNew The new royaltySupport address
    event UpdatedUtilityAndRoyaltyAddress(
        address dutchUtilityNew,
        address royaltySupporNew
    );

    // The address of the RoyaltyEngine to use via this contract
    IRoyaltyEngine public royaltySupport;

    // The address of the DutchUtility to use via this contract
    IDutchAuction public dutchUtility;

    // interface ID constants
    bytes4 private constant ERC721_INTERFACE_ID = 0x80ac58cd;
    bytes4 private constant ERC1155_INTERFACE_ID = 0xd9b67a26;

    // to validate auction Id
    mapping(string => bool) public usedAuctionId;

    // to list the auction details with respect to the auction id
    mapping(string => IDutchAuction.DutchAuctionList) private listings;

    // to set the buyers address with respect to the auction id
    mapping(string => EnumerableSet.AddressSet) private buyerAddress;

    // to set the sale details with respect to the auction id
    mapping(string => LatestSaleDetails) private latestSaleInfo;

    // to set the  buyerInfo with respect to the auction id and buyer
    mapping(string => mapping(address => BuyerDetails)) private buyerInfo;

    // to validate discount signature
    mapping(bytes => bool) public discountUsed;

    constructor(IRoyaltyEngine _royaltySupport, address dutchUtilityArg) {
        require(dutchUtilityArg != address(0), "invalid dutchUtilityArg");
        royaltySupport = _royaltySupport;
        dutchUtility = IDutchAuction(dutchUtilityArg);
    }

    /**
     * @notice createOrUpdateDutchAuction, creates and updates an auction
     * @param list the listing details to create an auction
     * @param dutchAuctionID the id of the auction
     */
    function createOrUpdateDutchAuction(
        string memory dutchAuctionID,
        IDutchAuction.DutchAuctionList memory list
    ) external nonReentrant {
        // checks caller: to be only called by the tokenOwner or admin
        require(
            isAdmin(msg.sender) || list.collectionList.tokenOwner == msg.sender,
            "allowed: only admin or token owner"
        );

        //calls the error handelling functionalities on utility contract
        dutchUtility.__beforeCreateAuction(list);

        // calls if the auction is created or updated
        if (!usedAuctionId[dutchAuctionID]) {
            listings[dutchAuctionID] = list;

            usedAuctionId[dutchAuctionID] = true;

            emit AuctionCreatedorUpdated(dutchAuctionID, list, "listed");
        } else if (usedAuctionId[dutchAuctionID]) {
            require(list.auctionStartTime > block.timestamp, "sale started");
            listings[dutchAuctionID] = list;
            emit AuctionCreatedorUpdated(dutchAuctionID, list, "updated");
        }
    }

    /**
     * @notice buy, buying the token(s) at current dutch auction price
     * @param dutchAuctionID the id of the created auction
     * @param price the buy amount to purchase NFT(s)
     * @param tax the tax amount of the purchase
     * @param quantity the number of tokens to be purchased
     * @param unit1155 the quantity for 1155
     * @param blacklistedProof the merkle tree path of buyer address
     * @param discount the discount related information
     */
    function buy(
        string memory dutchAuctionID,
        uint256 price,
        uint256 tax,
        uint256 quantity,
        uint256 unit1155,
        bytes32[] memory blacklistedProof,
        IDutchAuction.Discount memory discount
    ) external payable nonReentrant {
        // checks for auctionId is created for auction
        require(usedAuctionId[dutchAuctionID], "unsupported auction");

        IDutchAuction.NftContractList memory nftData = listings[dutchAuctionID]
            .collectionList;

        // checks whether is paying the amount with required amount
        if (listings[dutchAuctionID].paymentCurrency == address(0)) {
            require(msg.value == price + tax, "insufficient ETH");
        }

        uint256 refundTax; // refund Tax
        uint256 tempAmount; // refund Amount

        // validates the purchase data before buying the NFT
        (tempAmount, refundTax) = dutchUtility.__buyValidation(
            dutchAuctionID,
            msg.sender,
            IDutchAuction.BuyList(
                price,
                tax,
                quantity,
                unit1155,
                blacklistedProof,
                discount
            )
        );

        // transfers the excess amount to buyer
        if (
            listings[dutchAuctionID].paymentCurrency == address(0) &&
            (tempAmount + refundTax > 0)
        ) {
            (bool success, ) = payable(msg.sender).call{
                value: (tempAmount + refundTax)
            }(new bytes(0));
            require(success, "txn failed");
        }

        // updates the actual price and tax based the current price.
        price = price - tempAmount;
        tax = tax - refundTax;

        // adds buyer to the sale
        if (!buyerAddress[dutchAuctionID].contains(msg.sender)) {
            buyerAddress[dutchAuctionID].add(msg.sender);
        }

        // adds discount signature
        if (nftData.isSignerRequired) {
            discountUsed[discount.signature] = true;
        } else if (discount.discountPercentage > 0) {
            discountUsed[discount.signature] = true;
        }

        uint256 buyQuantity = (
            nftData.saleType ==
                IDutchAuction.TypeOfSale.single_tokens_multiple_Quantity
                ? unit1155
                : quantity
        );

        // adds the buy details based on discount %
        if (
            nftData.isDiscountRequired && discount.discountPercentage == 10000
        ) {
            latestSaleInfo[dutchAuctionID].fullDiscountSupply += buyQuantity;
        } else if (
            nftData.isDiscountRequired && discount.discountPercentage > 0
        ) {
            buyerInfo[dutchAuctionID][msg.sender].priceDiscount.push(price);
            buyerInfo[dutchAuctionID][msg.sender].taxDiscount.push(tax);
            buyerInfo[dutchAuctionID][msg.sender].quantityDiscount.push(
                buyQuantity
            );
        } else if (
            discount.discountPercentage == 0 || !nftData.isDiscountRequired
        ) {
            latestSaleInfo[dutchAuctionID].zeroDiscountSupply += buyQuantity;
            buyerInfo[dutchAuctionID][msg.sender].price.push(price);
            buyerInfo[dutchAuctionID][msg.sender].tax.push(tax);
            buyerInfo[dutchAuctionID][msg.sender].quantity.push(buyQuantity);
        }    

        // updating the purchase quantity and currentPrice
        buyerInfo[dutchAuctionID][msg.sender].totalCap += buyQuantity;
        latestSaleInfo[dutchAuctionID].currentSupply += buyQuantity;

        // Current Price is Stored in TempAmount Variable.
        tempAmount = dutchUtility.getCurrentDutchPrice(dutchAuctionID);
        latestSaleInfo[dutchAuctionID].finalPrice = tempAmount;

        if (latestSaleInfo[dutchAuctionID].nextTokenId == 0) {
            latestSaleInfo[dutchAuctionID].nextTokenId = nftData.startTokenId;
        }

        // Payment Transfer
        if (listings[dutchAuctionID].paymentCurrency != address(0)) {
            if (nftData.isRebate) {
                IERC20(listings[dutchAuctionID].paymentCurrency)
                    .safeTransferFrom(msg.sender, address(this), price + tax);
            }
        }
        if (!nftData.isRebate) {
            paymentTransaction(
                dutchAuctionID,
                price,
                tax,
                listings[dutchAuctionID].paymentCurrency,
                msg.sender,
                false
            );
        }
        uint256[] memory tokenIds;

        // transfer or mint the NFT
        if (nftData.isInstantDeliver) {
            (
                tokenIds,
                latestSaleInfo[dutchAuctionID].nextTokenId
            ) = __tokenTransaction(
                dutchAuctionID,
                msg.sender,
                quantity,
                nftData.saleType ==
                    IDutchAuction.TypeOfSale.single_tokens_multiple_Quantity
                    ? unit1155
                    : nftData.unit1155
            );
        }

        emit BuyExecuted(
            dutchAuctionID,
            quantity,
            unit1155,
            tempAmount,
            price,
            tax,
            discount.nonce,
            msg.sender,
            tokenIds
        );
    }

    /**
     * @notice endAuction, ending the auction
     * @param dutchAuctionID the id of the created auction
     */
    function endAuction(
        string memory dutchAuctionID,
        bool isPriceAsReserve
    ) external nonReentrant {
        // checks for auctionId is created for auction
        require(usedAuctionId[dutchAuctionID], "unsupported auction");

        IDutchAuction.NftContractList memory nftData = listings[dutchAuctionID]
            .collectionList;
        // calculates the total price of the auction
        uint256 totalPrice = 0;
        // calculates the tax of the total auction
        uint256 totalTax = 0;

        // checks the caller to be only the token owner or admin
        require(
            isAdmin(msg.sender) || nftData.tokenOwner == msg.sender,
            "allowed: only admin or token owner"
        );
        // checks the auction is crossed the end time or all nft are sold out or reached the max id
        require(
            listings[dutchAuctionID].auctionEndTime <= block.timestamp ||
                latestSaleInfo[dutchAuctionID].currentSupply ==
                (
                    nftData.saleType ==
                        IDutchAuction.TypeOfSale.single_tokens_multiple_Quantity
                        ? nftData.unit1155
                        : nftData.noOfTokens
                ) ||
                latestSaleInfo[dutchAuctionID].isReachedMaxId,
            "can't end"
        );
        // checks whether the auction is already ended
        require(!latestSaleInfo[dutchAuctionID].auctionEnded, "already ended");
        // checks for any bid in the provided auction
        require(latestSaleInfo[dutchAuctionID].currentSupply > 0, "no bids");
        // Sets Reserve price as final price.
        if (isPriceAsReserve) {
            latestSaleInfo[dutchAuctionID].finalPrice = listings[dutchAuctionID]
                .reservedPrice;
        }

        // checks the contract has sufficient balance
        if (nftData.isRebate) {
            (totalPrice, totalTax) = dutchUtility.computeSettlementAmountAndTax(
                dutchAuctionID
            );

            // transferring the payment to the represented address
            paymentTransaction(
                dutchAuctionID,
                totalPrice,
                totalTax,
                listings[dutchAuctionID].paymentCurrency,
                msg.sender,
                true
            );
        }

        uint256[] memory tokenIds;
        uint256[] memory totalTokenIds = new uint256[](
            latestSaleInfo[dutchAuctionID].currentSupply
        );
        // transferring the nft to the represented buyers
        if (!nftData.isInstantDeliver) {
            uint256 count = 0;
            for (
                uint256 i = 0;
                i < buyerAddress[dutchAuctionID].length();
                i++
            ) {
                address buyerAdddress = buyerAddress[dutchAuctionID].at(i);
                uint256 totalCap = buyerInfo[dutchAuctionID][buyerAdddress]
                    .totalCap;
                uint256 tokenCount = nftData.saleType ==
                    IDutchAuction.TypeOfSale.single_tokens_multiple_Quantity
                    ? 1
                    : totalCap;
                uint256 quantity = nftData.saleType ==
                    IDutchAuction.TypeOfSale.single_tokens_multiple_Quantity
                    ? totalCap
                    : nftData.unit1155;

                (
                    tokenIds,
                    latestSaleInfo[dutchAuctionID].nextTokenId
                ) = __tokenTransaction(
                    dutchAuctionID,
                    buyerAdddress,
                    tokenCount,
                    quantity
                );

                for (uint256 j = 0; j < tokenIds.length; j++) {
                    totalTokenIds[count] = tokenIds[j];
                    count++;
                }
            }
        }

        emit AuctionEnded(
            dutchAuctionID,
            latestSaleInfo[dutchAuctionID].currentSupply,
            latestSaleInfo[dutchAuctionID].finalPrice,
            totalTokenIds
        );
        latestSaleInfo[dutchAuctionID].auctionEnded = true;
    }

    /**
     * @notice cancelAuction, cancels the auction
     * @param auctionId the auction id to cancel
     */
    function cancelAuction(string memory auctionId) external adminRequired {
        require(usedAuctionId[auctionId], "unsupported auction");
        require(!latestSaleInfo[auctionId].auctionEnded, "already ended");
        require(latestSaleInfo[auctionId].finalPrice == 0, "sale started");
        delete (listings[auctionId]);
        emit CancelAuction(auctionId);
    }

    /**
     * @notice rebate, transferring the excess amount to the buyer
     * @param dutchAuctionID the id of the created auction
     * @param buyer the buyer wallet address
     */
    function rebate(
        string memory dutchAuctionID,
        address buyer
    ) external nonReentrant {
        // checks for auctionId is created for auction
        require(usedAuctionId[dutchAuctionID], "unsupported auction");
        // Checks for buyer is a part of buyers list in the auction
        require(
            buyerAddress[dutchAuctionID].contains(buyer),
            "invalid address"
        );
        // checks whether the auction is ended or sold out all NFTs
        require(
            latestSaleInfo[dutchAuctionID].auctionEnded ||
                latestSaleInfo[dutchAuctionID].currentSupply ==
                (
                    listings[dutchAuctionID].collectionList.saleType ==
                        IDutchAuction.TypeOfSale.single_tokens_multiple_Quantity
                        ? listings[dutchAuctionID].collectionList.unit1155
                        : listings[dutchAuctionID].collectionList.noOfTokens
                ),
            "not ended"
        );
        // checks whether the rebate is already claimed
        require(
            !buyerInfo[dutchAuctionID][buyer].rebateClaimed,
            "already claimed"
        );

        uint256 rebateAmount;
        uint256 rebateTax;

        // gets the rebate amount and tax from claimable()
        (rebateAmount, rebateTax) = dutchUtility.claimable(
            dutchAuctionID,
            buyer
        );
        // checks the rebate amount is greater than zero
        require(rebateAmount > 0, "no claim");

        emit Rebated(
            dutchAuctionID,
            buyer,
            rebateAmount,
            rebateTax,
            listings[dutchAuctionID].paymentCurrency
        );

        buyerInfo[dutchAuctionID][buyer].rebateClaimed = true;

        // transfers the payment to the buyer address
        _handlePayment(
            msg.sender,
            payable(buyer),
            listings[dutchAuctionID].paymentCurrency,
            rebateAmount + rebateTax,
            true
        );
    }

    /**
     * @notice Settles the payment to all settlement addresses concerned.
     * @param _dutchAuctionID the id of the auction
     * @param _totalAmount the totalAmount to be paid by the seller
     * @param tax the tax to be paid
     * @param _paymentToken the address of payment Token
     * @param _from the from address
     * @param isRebate the transaction status for rebate or not
     */
    function paymentTransaction(
        string memory _dutchAuctionID,
        uint256 _totalAmount,
        uint256 tax,
        address _paymentToken,
        address _from,
        bool isRebate
    ) private {
        IDutchAuction.SettlementList memory settlement = listings[
            _dutchAuctionID
        ].payoutList;

        uint256 totalCommession;

        _handlePayment(
            _from,
            settlement.taxSettlementAddress,
            _paymentToken,
            tax,
            isRebate
        );

        // transfers the platformFee amount to the platformSettlementAddress
        if (
            settlement.platformSettlementAddress != address(0) &&
            settlement.platformFeePercentage > 0
        ) {
            _handlePayment(
                _from,
                settlement.platformSettlementAddress,
                _paymentToken,
                totalCommession += ((_totalAmount *
                    settlement.platformFeePercentage) / 10000),
                isRebate
            );
        }

        // transfers the commissionfee amount to the commissionAddress
        if (
            settlement.commissionAddress != address(0) &&
            settlement.commissionFeePercentage > 0
        ) {
            totalCommession += ((_totalAmount *
                settlement.commissionFeePercentage) / 10000);
            _handlePayment(
                _from,
                settlement.commissionAddress,
                _paymentToken,
                ((_totalAmount * settlement.commissionFeePercentage) / 10000),
                isRebate
            );
        }
        _totalAmount = _totalAmount - totalCommession;

        // royalty fee payout settlement
        if (royaltySupport != IRoyaltyEngine(address(0))) {
            (
                address payable[] memory recipients,
                uint256[] memory bps // Royalty amount denominated in basis points
            ) = royaltySupport.getCollectionRoyalty(
                    listings[_dutchAuctionID].collectionList.nftContractAddress
                );

            // payouts each royalty
            for (uint256 i = 0; i < recipients.length; ) {
                uint256 feeAmount;

                feeAmount = (bps[i] * _totalAmount) / 10000;

                // ensures that we aren't somehow paying out more than we have
                require(_totalAmount >= feeAmount, "insolvent");

                emit RoyaltyPayout(
                    listings[_dutchAuctionID].collectionList.nftContractAddress,
                    recipients[i],
                    feeAmount
                );

                _handlePayment(
                    msg.sender,
                    recipients[i],
                    _paymentToken,
                    feeAmount,
                    isRebate
                );
                unchecked {
                    _totalAmount -= feeAmount;
                    ++i;
                }
            }
        }

        // transfers the balance to the paymentSettlementAddress
        _handlePayment(
            _from,
            settlement.paymentSettlementAddress,
            _paymentToken,
            _totalAmount,
            isRebate
        );
    }

    /**
     * @notice Settles the payment based on the given parameters
     * @param _to the address to whom need to settle the payment
     * @param _paymentToken the address of the ERC20 Payment Token
     * @param _amount the amount to be transferred
     */
    function _handlePayment(
        address _from,
        address payable _to,
        address _paymentToken,
        uint256 _amount,
        bool isRebate
    ) private {
        bool success;
        if (_paymentToken == address(0)) {
            // transfers the native currency
            (success, ) = _to.call{value: _amount}(new bytes(0));
            require(success, "txn failed");
        } else {
            if (isRebate) {
                // transfers the ERC20 currency from contract
                IERC20(_paymentToken).safeTransfer(_to, _amount);
            } else if (!isRebate) {
                // transfers the ERC20 currency from buyer wallet
                IERC20(_paymentToken).safeTransferFrom(_from, _to, _amount);
            }
        }
    }

    /**
     * @notice Transfers the nfts for erc-721/1155 collection
     * @param dutchAuctionID the id of created auction
     * @param buyer the address of the nft to be minted or transferred
     * @param quantity the number of tokens to be minted or transferred
     * @param unit1155 the units of tokens if erc-1155 collection
     */
    function __tokenTransaction(
        string memory dutchAuctionID,
        address buyer,
        uint256 quantity,
        uint256 unit1155
    ) private returns (uint256[] memory buyTokenIds, uint256 tokenId) {
        buyTokenIds = new uint256[](quantity);
        tokenId = latestSaleInfo[dutchAuctionID].nextTokenId;
        IDutchAuction.NftContractList memory nftData = listings[dutchAuctionID]
            .collectionList;
        // transfers the token.
        if (!nftData.isMint) {
            for (uint256 i = 0; i < quantity; i++) {
                if (
                    IERC165(nftData.nftContractAddress).supportsInterface(
                        ERC721_INTERFACE_ID
                    )
                ) {
                    while (
                        IERC721(nftData.nftContractAddress).ownerOf(tokenId) !=
                        nftData.tokenOwner
                    ) {
                        require(
                            nftData.endTokenId >= tokenId,
                            "not enough token"
                        );
                        tokenId += 1;
                    }
                    // Transfers the ERC721 token(s)
                    IERC721(nftData.nftContractAddress).safeTransferFrom(
                        nftData.tokenOwner,
                        buyer,
                        tokenId
                    );
                    buyTokenIds[i] = tokenId;
                    tokenId += 1;
                } else if (
                    IERC165(nftData.nftContractAddress).supportsInterface(
                        ERC1155_INTERFACE_ID
                    )
                ) {
                    if (
                        nftData.saleType ==
                        IDutchAuction
                            .TypeOfSale
                            .multiple_tokens_with_Same_Quantity
                    ) {
                        while (
                            IERC1155(nftData.nftContractAddress).balanceOf(
                                nftData.tokenOwner,
                                tokenId
                            ) < unit1155
                        ) {
                            require(
                                nftData.endTokenId >= tokenId,
                                "not enough token"
                            );
                            tokenId += 1;
                        }

                        // Transfers the ERC1155 Token(s)
                        IERC1155(nftData.nftContractAddress).safeTransferFrom(
                            nftData.tokenOwner,
                            buyer,
                            tokenId,
                            unit1155,
                            "0x"
                        );
                        buyTokenIds[i] = tokenId;
                        tokenId += 1;
                    } else if (
                        nftData.saleType ==
                        IDutchAuction.TypeOfSale.single_tokens_multiple_Quantity
                    ) {
                        require(
                            IERC1155(nftData.nftContractAddress).balanceOf(
                                nftData.tokenOwner,
                                tokenId
                            ) > unit1155,
                            "insufficient quantity"
                        );
                        // transferring erc-1155
                        IERC1155(nftData.nftContractAddress).safeTransferFrom(
                            nftData.tokenOwner,
                            buyer,
                            tokenId,
                            unit1155,
                            "0x"
                        );
                        buyTokenIds[i] = tokenId;
                    }
                }
            }
            return (buyTokenIds, tokenId);
            // mints the NFT to buyer address
        } else if (nftData.isMint) {
            if (
                IERC165(nftData.nftContractAddress).supportsInterface(
                    ERC721_INTERFACE_ID
                )
            ) {
                // mints for erc-721 collection
                buyTokenIds = IERC721CreatorCore(nftData.nftContractAddress)
                    .mintExtensionBatch(buyer, uint16(quantity));
            } else if (
                IERC165(nftData.nftContractAddress).supportsInterface(
                    ERC1155_INTERFACE_ID
                )
            ) {
                address[] memory to = new address[](1);
                uint256[] memory amounts = new uint256[](quantity);
                string[] memory uris;
                to[0] = buyer;

                if (
                    nftData.saleType ==
                    IDutchAuction.TypeOfSale.multiple_tokens_with_Same_Quantity
                ) {
                    for (uint256 i = 0; i < quantity; i++) {
                        amounts[i] = unit1155;
                    }
                    // mints nft for erc-1155 collections for new nfts
                    buyTokenIds = IERC1155CreatorCore(
                        nftData.nftContractAddress
                    ).mintExtensionNew(to, amounts, uris);
                } else if (
                    nftData.saleType ==
                    IDutchAuction.TypeOfSale.single_tokens_multiple_Quantity
                ) {
                    amounts[0] = unit1155;
                    if (tokenId == 0) {
                        // mints nft for erc-1155 collections as new nfts
                        buyTokenIds = IERC1155CreatorCore(
                            nftData.nftContractAddress
                        ).mintExtensionNew(to, amounts, uris);
                        latestSaleInfo[dutchAuctionID]
                            .nextTokenId = buyTokenIds[0];
                    } else {
                        // mints nft for erc-1155 collections as existing nfts
                        buyTokenIds[0] = tokenId;
                        IERC1155CreatorCore(nftData.nftContractAddress)
                            .mintExtensionExisting(to, buyTokenIds, amounts);
                    }
                }
            }
            if (nftData.maxTokenIDRange != -1) {
                if (
                    buyTokenIds[buyTokenIds.length - 1] >=
                    uint256(nftData.maxTokenIDRange)
                ) {
                    latestSaleInfo[dutchAuctionID].isReachedMaxId = true;
                }
                require(
                    buyTokenIds[buyTokenIds.length - 1] <=
                        uint256(nftData.maxTokenIDRange),
                    "reached max id"
                );
            }

            return (buyTokenIds, buyTokenIds[buyTokenIds.length - 1]);
        }
    }

    /**
     * @notice Withdraw the funds to owner
     * @param paymentCurrency the address of the paymentCurrency
     */
    function withdraw(address paymentCurrency) external onlyOwner {
        bool success;
        if (paymentCurrency == address(0)) {
            (success, ) = payable(msg.sender).call{
                value: address(this).balance
            }(new bytes(0));
            require(success, "txn failed");
        } else if (paymentCurrency != address(0)) {
            // transfers the ERC20 currency
            IERC20(paymentCurrency).safeTransfer(
                payable(msg.sender),
                IERC20(paymentCurrency).balanceOf(address(this))
            );
        }
    }

    /**
     * @notice Updates the end time of auction
     * @param dutchAuctionID the id of created auction
     * @param endTime the end time to update
     */
    function updateEndTime(
        string memory dutchAuctionID,
        uint32 endTime
    ) external adminRequired {
        require(usedAuctionId[dutchAuctionID], "unsupported auction");
        // checks whether the auction is already ended
        require(!latestSaleInfo[dutchAuctionID].auctionEnded, "already ended");
        emit EndTimeUpdated(dutchAuctionID, endTime);
        listings[dutchAuctionID].auctionEndTime = endTime;
    }

    /**
     * @notice Updates the state variables in contract
     * @param dutchAuctionID the id of created Auction
     * @param reservedPrice the reserved price to update
     */
    function updateReservePrice(
        string memory dutchAuctionID,
        uint128 reservedPrice
    ) external adminRequired {
        require(usedAuctionId[dutchAuctionID], "unsupported auction");
        require(
            reservedPrice < dutchUtility.getCurrentDutchPrice(dutchAuctionID),
            "invalid arg"
        );
        emit ReservePriceUpdated(dutchAuctionID, reservedPrice);
        listings[dutchAuctionID].reservedPrice = reservedPrice;
    }

    /**
     * @notice Updates the state variables in contract
     * @param royaltySupportArg the address for royalty payment support
     * @param dutchUtilityArg the address for utility support
     */
    function updateUtilityAndRoyaltyAddress(
        address royaltySupportArg,
        address dutchUtilityArg
    ) external adminRequired {
        require(dutchUtilityArg != address(0), "Invalid dutchUtilityArg");
        emit UpdatedUtilityAndRoyaltyAddress(
            dutchUtilityArg,
            royaltySupportArg
        );
        royaltySupport = IRoyaltyEngine(royaltySupportArg);
        dutchUtility = IDutchAuction(dutchUtilityArg);
    }

    /**
     * @notice Gets the detils of the listed sale
     * @param dutchAuctionID the id of the created auction
     */
    function getListings(
        string memory dutchAuctionID
    ) external view returns (IDutchAuction.DutchAuctionList memory auctionist) {
        return listings[dutchAuctionID];
    }

    /**
     * @notice Gets the detils collection details of listed sale
     * @param dutchAuctionID the id of the created auction
     */
    function getLatestSaleInfo(
        string memory dutchAuctionID
    ) external view returns (LatestSaleDetails memory saleDetails) {
        return latestSaleInfo[dutchAuctionID];
    }

    /**
     * @notice Gets the buyers sale info of the listed sale
     * @param dutchAuctionID the id of the created auction
     * @param buyer the address of the buyer
     */
    function getBuyerInfo(
        string memory dutchAuctionID,
        address buyer
    ) external view returns (BuyerDetails memory buyerDetails) {
        return buyerInfo[dutchAuctionID][buyer];
    }

    /**
     * @notice Gets the list of all the buyers in the sale
     * @param dutchAuctionID the id of the created auction
     */
    function getBuyerAddress(
        string memory dutchAuctionID
    ) external view returns (address[] memory buyers) {
        return buyerAddress[dutchAuctionID].values();
    }
}

// File: contracts/openzeppelin-upgradeable/utils/AddressUpgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (utils/Address.sol)

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library AddressUpgradeable {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: contracts/openzeppelin/access/Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../utils/Context.sol";
/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor () {
        address msgSender = _msgSender();
        _owner = msgSender;
        emit OwnershipTransferred(address(0), msgSender);
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        emit OwnershipTransferred(_owner, address(0));
        _owner = address(0);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        emit OwnershipTransferred(_owner, newOwner);
        _owner = newOwner;
    }
}

// File: contracts/openzeppelin/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    /// @custom:oz-upgrades-unsafe-allow constructor
    constructor () {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and make it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: contracts/openzeppelin/token/ERC1155/IERC1155.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (token/ERC1155/IERC1155.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC1155 compliant contract, as defined in the
 * https://eips.ethereum.org/EIPS/eip-1155[EIP].
 *
 * _Available since v3.1._
 */
interface IERC1155 is IERC165 {
    /**
     * @dev Emitted when `value` tokens of token type `id` are transferred from `from` to `to` by `operator`.
     */
    event TransferSingle(address indexed operator, address indexed from, address indexed to, uint256 id, uint256 value);

    /**
     * @dev Equivalent to multiple {TransferSingle} events, where `operator`, `from` and `to` are the same for all
     * transfers.
     */
    event TransferBatch(
        address indexed operator,
        address indexed from,
        address indexed to,
        uint256[] ids,
        uint256[] values
    );

    /**
     * @dev Emitted when `account` grants or revokes permission to `operator` to transfer their tokens, according to
     * `approved`.
     */
    event ApprovalForAll(address indexed account, address indexed operator, bool approved);

    /**
     * @dev Emitted when the URI for token type `id` changes to `value`, if it is a non-programmatic URI.
     *
     * If an {URI} event was emitted for `id`, the standard
     * https://eips.ethereum.org/EIPS/eip-1155#metadata-extensions[guarantees] that `value` will equal the value
     * returned by {IERC1155MetadataURI-uri}.
     */
    event URI(string value, uint256 indexed id);

    /**
     * @dev Returns the amount of tokens of token type `id` owned by `account`.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function balanceOf(address account, uint256 id) external view returns (uint256);

    /**
     * @dev xref:ROOT:erc1155.adoc#batch-operations[Batched] version of {balanceOf}.
     *
     * Requirements:
     *
     * - `accounts` and `ids` must have the same length.
     */
    function balanceOfBatch(address[] calldata accounts, uint256[] calldata ids)
        external
        view
        returns (uint256[] memory);

    /**
     * @dev Grants or revokes permission to `operator` to transfer the caller's tokens, according to `approved`,
     *
     * Emits an {ApprovalForAll} event.
     *
     * Requirements:
     *
     * - `operator` cannot be the caller.
     */
    function setApprovalForAll(address operator, bool approved) external;

    /**
     * @dev Returns true if `operator` is approved to transfer ``account``'s tokens.
     *
     * See {setApprovalForAll}.
     */
    function isApprovedForAll(address account, address operator) external view returns (bool);

    /**
     * @dev Transfers `amount` tokens of token type `id` from `from` to `to`.
     *
     * Emits a {TransferSingle} event.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - If the caller is not `from`, it must be have been approved to spend ``from``'s tokens via {setApprovalForAll}.
     * - `from` must have a balance of tokens of type `id` of at least `amount`.
     * - If `to` refers to a smart contract, it must implement {IERC1155Receiver-onERC1155Received} and return the
     * acceptance magic value.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 id,
        uint256 amount,
        bytes calldata data
    ) external;

    /**
     * @dev xref:ROOT:erc1155.adoc#batch-operations[Batched] version of {safeTransferFrom}.
     *
     * Emits a {TransferBatch} event.
     *
     * Requirements:
     *
     * - `ids` and `amounts` must have the same length.
     * - If `to` refers to a smart contract, it must implement {IERC1155Receiver-onERC1155BatchReceived} and return the
     * acceptance magic value.
     */
    function safeBatchTransferFrom(
        address from,
        address to,
        uint256[] calldata ids,
        uint256[] calldata amounts,
        bytes calldata data
    ) external;
}

// File: contracts/openzeppelin/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: contracts/openzeppelin/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: contracts/openzeppelin/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}

// File: contracts/openzeppelin/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (utils/Address.sol)

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: contracts/openzeppelin/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/openzeppelin/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: contracts/openzeppelin/utils/introspection/ERC165Checker.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (utils/introspection/ERC165Checker.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Library used to query support of an interface declared via {IERC165}.
 *
 * Note that these functions return the actual result of the query: they do not
 * `revert` if an interface is not supported. It is up to the caller to decide
 * what to do in these cases.
 */
library ERC165Checker {
    // As per the EIP-165 spec, no interface should ever match 0xffffffff
    bytes4 private constant _INTERFACE_ID_INVALID = 0xffffffff;

    /**
     * @dev Returns true if `account` supports the {IERC165} interface,
     */
    function supportsERC165(address account) internal view returns (bool) {
        // Any contract that implements ERC165 must explicitly indicate support of
        // InterfaceId_ERC165 and explicitly indicate non-support of InterfaceId_Invalid
        return
            _supportsERC165Interface(account, type(IERC165).interfaceId) &&
            !_supportsERC165Interface(account, _INTERFACE_ID_INVALID);
    }

    /**
     * @dev Returns true if `account` supports the interface defined by
     * `interfaceId`. Support for {IERC165} itself is queried automatically.
     *
     * See {IERC165-supportsInterface}.
     */
    function supportsInterface(address account, bytes4 interfaceId) internal view returns (bool) {
        // query support of both ERC165 as per the spec and support of _interfaceId
        return supportsERC165(account) && _supportsERC165Interface(account, interfaceId);
    }

    /**
     * @dev Returns a boolean array where each value corresponds to the
     * interfaces passed in and whether they're supported or not. This allows
     * you to batch check interfaces for a contract where your expectation
     * is that some interfaces may not be supported.
     *
     * See {IERC165-supportsInterface}.
     *
     * _Available since v3.4._
     */
    function getSupportedInterfaces(address account, bytes4[] memory interfaceIds)
        internal
        view
        returns (bool[] memory)
    {
        // an array of booleans corresponding to interfaceIds and whether they're supported or not
        bool[] memory interfaceIdsSupported = new bool[](interfaceIds.length);

        // query support of ERC165 itself
        if (supportsERC165(account)) {
            // query support of each interface in interfaceIds
            for (uint256 i = 0; i < interfaceIds.length; i++) {
                interfaceIdsSupported[i] = _supportsERC165Interface(account, interfaceIds[i]);
            }
        }

        return interfaceIdsSupported;
    }

    /**
     * @dev Returns true if `account` supports all the interfaces defined in
     * `interfaceIds`. Support for {IERC165} itself is queried automatically.
     *
     * Batch-querying can lead to gas savings by skipping repeated checks for
     * {IERC165} support.
     *
     * See {IERC165-supportsInterface}.
     */
    function supportsAllInterfaces(address account, bytes4[] memory interfaceIds) internal view returns (bool) {
        // query support of ERC165 itself
        if (!supportsERC165(account)) {
            return false;
        }

        // query support of each interface in _interfaceIds
        for (uint256 i = 0; i < interfaceIds.length; i++) {
            if (!_supportsERC165Interface(account, interfaceIds[i])) {
                return false;
            }
        }

        // all interfaces supported
        return true;
    }

    /**
     * @notice Query if a contract implements an interface, does not check ERC165 support
     * @param account The address of the contract to query for support of an interface
     * @param interfaceId The interface identifier, as specified in ERC-165
     * @return true if the contract at account indicates support of the interface with
     * identifier interfaceId, false otherwise
     * @dev Assumes that account contains a contract that supports ERC165, otherwise
     * the behavior of this method is undefined. This precondition can be checked
     * with {supportsERC165}.
     * Interface identification is specified in ERC-165.
     */
    function _supportsERC165Interface(address account, bytes4 interfaceId) private view returns (bool) {
        bytes memory encodedParams = abi.encodeWithSelector(IERC165.supportsInterface.selector, interfaceId);
        (bool success, bytes memory result) = account.staticcall{gas: 30000}(encodedParams);
        if (result.length < 32) return false;
        return success && abi.decode(result, (bool));
    }
}

// File: contracts/openzeppelin/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: contracts/openzeppelin/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: contracts/openzeppelin/utils/structs/EnumerableSet.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (utils/structs/EnumerableSet.sol)

pragma solidity ^0.8.0;

/**
 * @dev Library for managing
 * https://en.wikipedia.org/wiki/Set_(abstract_data_type)[sets] of primitive
 * types.
 *
 * Sets have the following properties:
 *
 * - Elements are added, removed, and checked for existence in constant time
 * (O(1)).
 * - Elements are enumerated in O(n). No guarantees are made on the ordering.
 *
 * ```
 * contract Example {
 *     // Add the library methods
 *     using EnumerableSet for EnumerableSet.AddressSet;
 *
 *     // Declare a set state variable
 *     EnumerableSet.AddressSet private mySet;
 * }
 * ```
 *
 * As of v3.3.0, sets of type `bytes32` (`Bytes32Set`), `address` (`AddressSet`)
 * and `uint256` (`UintSet`) are supported.
 */
library EnumerableSet {
    // To implement this library for multiple types with as little code
    // repetition as possible, we write it in terms of a generic Set type with
    // bytes32 values.
    // The Set implementation uses private functions, and user-facing
    // implementations (such as AddressSet) are just wrappers around the
    // underlying Set.
    // This means that we can only create new EnumerableSets for types that fit
    // in bytes32.

    struct Set {
        // Storage of set values
        bytes32[] _values;
        // Position of the value in the `values` array, plus 1 because index 0
        // means a value is not in the set.
        mapping(bytes32 => uint256) _indexes;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function _add(Set storage set, bytes32 value) private returns (bool) {
        if (!_contains(set, value)) {
            set._values.push(value);
            // The value is stored at length-1, but we add 1 to all indexes
            // and use 0 as a sentinel value
            set._indexes[value] = set._values.length;
            return true;
        } else {
            return false;
        }
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function _remove(Set storage set, bytes32 value) private returns (bool) {
        // We read and store the value's index to prevent multiple reads from the same storage slot
        uint256 valueIndex = set._indexes[value];

        if (valueIndex != 0) {
            // Equivalent to contains(set, value)
            // To delete an element from the _values array in O(1), we swap the element to delete with the last one in
            // the array, and then remove the last element (sometimes called as 'swap and pop').
            // This modifies the order of the array, as noted in {at}.

            uint256 toDeleteIndex = valueIndex - 1;
            uint256 lastIndex = set._values.length - 1;

            if (lastIndex != toDeleteIndex) {
                bytes32 lastvalue = set._values[lastIndex];

                // Move the last value to the index where the value to delete is
                set._values[toDeleteIndex] = lastvalue;
                // Update the index for the moved value
                set._indexes[lastvalue] = valueIndex; // Replace lastvalue's index to valueIndex
            }

            // Delete the slot where the moved value was stored
            set._values.pop();

            // Delete the index for the deleted slot
            delete set._indexes[value];

            return true;
        } else {
            return false;
        }
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function _contains(Set storage set, bytes32 value) private view returns (bool) {
        return set._indexes[value] != 0;
    }

    /**
     * @dev Returns the number of values on the set. O(1).
     */
    function _length(Set storage set) private view returns (uint256) {
        return set._values.length;
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function _at(Set storage set, uint256 index) private view returns (bytes32) {
        return set._values[index];
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function _values(Set storage set) private view returns (bytes32[] memory) {
        return set._values;
    }

    // Bytes32Set

    struct Bytes32Set {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(Bytes32Set storage set, bytes32 value) internal returns (bool) {
        return _add(set._inner, value);
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(Bytes32Set storage set, bytes32 value) internal returns (bool) {
        return _remove(set._inner, value);
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(Bytes32Set storage set, bytes32 value) internal view returns (bool) {
        return _contains(set._inner, value);
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(Bytes32Set storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(Bytes32Set storage set, uint256 index) internal view returns (bytes32) {
        return _at(set._inner, index);
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(Bytes32Set storage set) internal view returns (bytes32[] memory) {
        return _values(set._inner);
    }

    // AddressSet

    struct AddressSet {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(AddressSet storage set, address value) internal returns (bool) {
        return _add(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(AddressSet storage set, address value) internal returns (bool) {
        return _remove(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(AddressSet storage set, address value) internal view returns (bool) {
        return _contains(set._inner, bytes32(uint256(uint160(value))));
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(AddressSet storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(AddressSet storage set, uint256 index) internal view returns (address) {
        return address(uint160(uint256(_at(set._inner, index))));
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(AddressSet storage set) internal view returns (address[] memory) {
        bytes32[] memory store = _values(set._inner);
        address[] memory result;

        assembly {
            result := store
        }

        return result;
    }

    // UintSet

    struct UintSet {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(UintSet storage set, uint256 value) internal returns (bool) {
        return _add(set._inner, bytes32(value));
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(UintSet storage set, uint256 value) internal returns (bool) {
        return _remove(set._inner, bytes32(value));
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(UintSet storage set, uint256 value) internal view returns (bool) {
        return _contains(set._inner, bytes32(value));
    }

    /**
     * @dev Returns the number of values on the set. O(1).
     */
    function length(UintSet storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

    /**
     * @dev Returns the value stored at position `index` in the set. O(1).
     *
     * Note that there are no guarantees on the ordering of values inside the
     * array, and it may change when more values are added or removed.
     *
     * Requirements:
     *
     * - `index` must be strictly less than {length}.
     */
    function at(UintSet storage set, uint256 index) internal view returns (uint256) {
        return uint256(_at(set._inner, index));
    }

    /**
     * @dev Return the entire set in an array
     *
     * WARNING: This operation will copy the entire storage to memory, which can be quite expensive. This is designed
     * to mostly be used by view accessors that are queried without any gas fees. Developers should keep in mind that
     * this function has an unbounded cost, and using it as part of a state-changing function may render the function
     * uncallable if the set grows to a point where copying to memory consumes too much gas to fit in a block.
     */
    function values(UintSet storage set) internal view returns (uint256[] memory) {
        bytes32[] memory store = _values(set._inner);
        uint256[] memory result;

        assembly {
            result := store
        }

        return result;
    }
}